












Шапка	1	<Банк>										БИК										<КодБанка>									
	2											Сч. №																			
	3																														
	4	УНП <УНП>										Сч. №										<НомерСчетаОрганизации>									
	5	<Организация>																													
	6																														
	7																														
	8																														
<Счет на оплату № [Номер] от [Дата]>																															
Поставщик (исполнитель):	13	<ПредставлениеОрганизации>																													
	14																														
	15																														
	16	<ПредставлениеКонтрагента>																													
Покупатель (заказчик):	17																														
	18																														
Основание:	19	<Договор [НомерДоговора] от [ДатаДоговора]>																													
ШапкаТаблиц	22	№	Товар (Услуга)										Кол-во	Цена	Сумма	Ставка НДС	Сумма НДС	Всего с НДС													
	23	1	2										3	4	5	6	7	8													
СтрокаТаблиц	24	мерСтр	<Номенклатура>										<Количество>	<Цена>	<Сумма>	<СтавкаНДС>	<СуммаНДС>	<СуммаВсего>													
Подвал	25													<ИтогСумма>		ИтогСуммаНДС	<ИтогСуммаВсего>														
	26																														
Подписи	27											Итого:										<ИтогСумма>									
	28											Сумма НДС:										<ИтогСуммаНДС>									
	29											Всего к оплате:										<ИтогСуммаВсего>									
	30																														
	31	Всего к оплате с НДС:																													
	32	<ИтогСуммаВсегоПрописью>																													
33																															
34																															
Руководитель	35											<Руководитель>																			
36											подпись										расшифровка подписи										
37																															

Вручную создаем макет, который нам потребуется для заполнения

- [-]  ЗаказКлиента
 - [+]  Реквизиты
 - [+]  Табличные части
 - [+]  Формы
 - [-]  Команды
 -  СчетНаОплату
 -  ВыгрузитьВФайл
 -  ПечатьСчетаНаОплату
 - [-]  Макеты
 -  СчетНаОплату
 -  ПФ_MXL_СчетНаОплату

Создал команду для формирования такого макета

```

&НаКлиенте
Процедура ОбработкаКоманды(ПараметрКоманды, ПараметрыВыполненияКоманды)

    //1. Создать новый табличный документ
    ТабДок = Новый ТабличныйДокумент;

    //2. Заполнить табличный документ, используя области из макета:
    ОбработкаКомандыНаСервере(ПараметрКоманды, ТабДок);

    //3. Настроить свойства табличного документа (ориентация страницы, отступы и т.д.)
    ТабДок.АвтоМасштаб = Истина;
    ТабДок.ОтображатьЗаголовки = Ложь;
    ТабДок.ОтображатьСетку = Ложь;

    //Показать печатную форму
    ТабДок.Показать();

    ////4. Показать печатную форму пользователя
    ////ТабДок.Показать();
    //СтруктураПараметров = Новый Структура("ТабДок", ТабДок);
    //ОткрытьФорму("Обработка.ПечатьДокументов.Форма.Форма", СтруктураПараметров);
    //
КонецПроцедуры


```

Создаем табличный документ и вызываем процедуру на сервере, поскольку нам потребуется заполнять ТабДок, а вызов данных производится на сервере. Далее производим настройку внешнего вида и корректировку масштаба получившегося документа и отображаем полученный документ.

```

&НаСервере
Процедура ОбработкаКомандыНаСервере(МассивДокументов, ТабДок)
    Документы.ЗаказКлиента.ПечатьСчетаНаОплату(ТабДок, МассивДокументов);
КонецПроцедуры

```



На этапе вызова серверной процедуры мы расположим код по выборке данных для заполнения в модуле менеджера, поскольку в дальнейшем может потребоваться вызывать команду не только в документе заказы клиента, но и в других местах.

Процедура ПечатьСчетаНаОплату(ТабДок, МассивДокументов, ВыводитьПечать=Ложь) Экспорт

```
//2. Получить макет печатной формы
Макет = Документы.ЗаказКлиента.ПолучитьМакет("Пф_MXL_СчетНаОплату");

//получение данных документов
Запрос = Новый Запрос;
Запрос.Текст =
"ВЫБРАТЬ
|   ЗаказКлиента.Номер КАК Номер,
|   ЗаказКлиента.Дата КАК Дата,
|   ЗаказКлиента.Организация КАК Организация,
|   ЗаказКлиента.Организация.ОсновнойБанковскийСчет.Банк КАК Банк,
|   ЗаказКлиента.Организация.ОсновнойБанковскийСчет.НомерСчета КАК НомерСчетаОрганизации,
|   ЗаказКлиента.Организация.ОсновнойБанковскийСчет.Банк.Код КАК КодБанка,
|   ЗаказКлиента.Договор.Номер КАК НомерДоговора,
|   ЗаказКлиента.Договор.ДатаДоговора КАК ДатаДоговора,
|   ЗаказКлиента.Организация.Руководитель КАК Руководитель,
|   ЗаказКлиента.Товары. {
|       НомерСтроки КАК НомерСтроки,
|       Номенклатура КАК Номенклатура,
|       Количество КАК Количество,
|       ЕдиницаИзмерения КАК ЕдиницаИзмерения,
|       Цена КАК Цена,
|       СтавкаНДС КАК СтавкаНДС,
|       Сумма КАК Сумма,
|       СуммаНДС КАК СуммаНДС,
|       СуммаВсего КАК СуммаВсего
|   } КАК Товары,
|   ЗаказКлиента.Организация.ИНН КАК УНП,
|   ЗаказКлиента.Контрагент КАК Контрагент,
|   ЗаказКлиента.Контрагент.ИНН КАК КонтрагентИНН,
|   ЕСТЬNULL(ОрганизацииКонтактнаяИнформация.Значение, "") КАК ЮрАдресОрганизации,
|   ЕСТЬNULL(КонтрагентыКонтактнаяИнформация.Значение, "") КАК ЮрАдресКонтрагента,
|   ЗаказКлиента.Организация.Логотип КАК Логотип,
|   ЗаказКлиента.Организация.Печать КАК ПечатьОрганизации,
|   ЕСТЬNULL(ЗаказКлиента.Организация.Руководитель.Подпись, НЕОПРЕДЕЛЕНО) КАК ПодписьРуководителя
| ИЗ
|   Документ.ЗаказКлиента КАК ЗаказКлиента
|       ЛЕВОЕ СОЕДИНЕНИЕ Справочник.Организации.КонтактнаяИнформация КАК ОрганизацииКонтактнаяИнформация
|       ПО ЗаказКлиента.Организация = ОрганизацииКонтактнаяИнформация.Ссылка
|           И (ОрганизацииКонтактнаяИнформация.Вид = &ВидКИ_ЮрАдресОрганизации)
|       ЛЕВОЕ СОЕДИНЕНИЕ Справочник.Контрагенты.КонтактнаяИнформация КАК КонтрагентыКонтактнаяИнформация
|       ПО ЗаказКлиента.Контрагент = КонтрагентыКонтактнаяИнформация.Ссылка
|           И (КонтрагентыКонтактнаяИнформация.Вид = &ВидКИ_ЮрАдресКонтрагента)

| ГДЕ
|   ЗаказКлиента.Ссылка В (&МассивДокументов)";

Запрос.УстановитьПараметр("МассивДокументов", МассивДокументов);
Запрос.УстановитьПараметр("ВидКИ_ЮрАдресОрганизации", Справочники.ВидыКонтактнойИнформации.ЮридическийАдресОрганизации);
Запрос.УстановитьПараметр("ВидКИ_ЮрАдресКонтрагента", Справочники.ВидыКонтактнойИнформации.ЮридическийАдресКонтрагента);

РезультатЗапроса = Запрос.Выполнить();

Выборка = РезультатЗапроса.Выбрать();

ПервыйДокумент = Истина;

Пока Выборка.Следующий() Цикл
    Если ПервыйДокумент Тогда
        ПервыйДокумент = Ложь;
    Иначе
        ТабДок.ВывестиГоризонтальныйРазделительСтраниц();
    КонецЕсли;

    //Получить область из макета
    Область = Макет.ПолучитьОбласть("Шапка");

    ХранилищеЛоготипа = Выборка.Логотип;
    ДвоичныеДанныеЛоготипа = ХранилищеЛоготипа.Получить();

    Если ДвоичныеДанныеЛоготипа <> Неопределено Тогда
        //вывод логотипа
        Область.Рисунки.Логотип.Картинка = Новый Картинка(ДвоичныеДанныеЛоготипа);
    КонецЕсли;

    //Заполнить параметры области (если они есть)
    СтруктураЗначений = Новый Структура;
    СтруктураЗначений.Вставить("ПредставлениеОрганизации", "" + Выборка.Организация + ", УНП " + Выборка.УНП + ", " + Выборка.ЮрАдресОрганизации);
    СтруктураЗначений.Вставить("ПредставлениеКонтрагента", "" + Выборка.Контрагент + ", УНП " + Выборка.КонтрагентИНН + ", " + Выборка.ЮрАдресКонтрагента);

    Область.Параметры.Заполнить(Выборка);
    Область.Параметры.Заполнить(СтруктураЗначений);

    //Вывести область в табличный документ
    ТабДок.Вывести(Область);

    Область = Макет.ПолучитьОбласть("ШапкаТаблицы");
    ТабДок.Вывести(Область);
```

```

СтруктураИтогов = Новый Структура("ИтогСумма, ИтогСуммаНДС, ИтогСуммаВсего, ИтогСуммаВсегоПрописью", 0, 0, 0);

ВыборкаТовары = Выборка.Товары.Выбрать();
Пока ВыборкаТовары.Следующий() Цикл

    Область = Макет.ПолучитьОбласть("СтрокаТаблицы");
    Область.Параметры.Заполнить(ВыборкаТовары);
    ТабДок.Вывести(Область);

    СтруктураИтогов.ИтогСумма = СтруктураИтогов.ИтогСумма + ВыборкаТовары.Сумма;
    СтруктураИтогов.ИтогСуммаНДС = СтруктураИтогов.ИтогСуммаНДС + ВыборкаТовары.СуммаНДС;
    СтруктураИтогов.ИтогСуммаВсего = СтруктураИтогов.ИтогСуммаВсего + ВыборкаТовары.СуммаВсего;

КонецЦикла;

ПараметрыПрописи = "рубль, рубля, рублей, м, копейка, копейки, копеек, ж, 2";
СтруктураИтогов.ИтогСуммаВсегоПрописью = ЧислоПрописью(СтруктураИтогов.ИтогСуммаВсего, "ДП=Истина", ПараметрыПрописи);

Область = Макет.ПолучитьОбласть("Подвал");
Область.Параметры.Заполнить(СтруктураИтогов);
ТабДок.Вывести(Область);

Если ВыводитьПечать Тогда

    БезЛинии = Новый Линия(ТипЛинииРисункаТабличногоДокумента.НетЛинии);

    Область = Макет.ПолучитьОбласть("ПодписьФаксимиле");
    Область.Параметры.Заполнить(СтруктураИтогов);

    //вывод подписи руководителя
    ХранилищеПодписи = Выборка.ПодписьРуководителя;
    Если ХранилищеПодписи <> Неопределено Тогда

        ДвоичныеДанныеПодписи = ХранилищеПодписи.Получить();

        Если ДвоичныеДанныеПодписи <> Неопределено Тогда
            //вывод подпись руководителя
            Область.Рисунки.ПодписьРуководителя.Картинка = Новый Картинка(ДвоичныеДанныеПодписи);
            Область.Рисунки.ПодписьРуководителя.Линия = БезЛинии;
        КонецЕсли;

    КонецЕсли;

    //вывод печати организации
    ХранилищеПечати = Выборка.ПечатьОрганизации;
    ДвоичныеДанныеПечати = ХранилищеПечати.Получить();

    Если ДвоичныеДанныеПечати <> Неопределено Тогда
        //вывод подпись руководителя
        Область.Рисунки.ПечатьОрганизации.Картинка = Новый Картинка(ДвоичныеДанныеПечати);
        Область.Рисунки.ПечатьОрганизации.Линия = БезЛинии;
    КонецЕсли;

    ТабДок.Вывести(Область);

Иначе
    Область = Макет.ПолучитьОбласть("Подписи");
    Область.Параметры.Заполнить(Выборка);
    ТабДок.Вывести(Область);
КонецЕсли;

КонецЦикла;

КонецПроцедуры

```

И далее в целом идет процесс заполнения макета